/****************************************************************************
**
** Copyright (C) 2015 The Qt Company Ltd.
** Contact: http://www.qt.io/licensing/
**
** This file is part of the examples of the Qt Toolkit.
**
** $QT_BEGIN_LICENSE:BSD$
** You may use this file under the terms of the BSD license as follows:
**
** "Redistribution and use in source and binary forms, with or without
** modification, are permitted provided that the following conditions are
** met:
**   * Redistributions of source code must retain the above copyright
**     notice, this list of conditions and the following disclaimer.
**   * Redistributions in binary form must reproduce the above copyright
**     notice, this list of conditions and the following disclaimer in
**     the documentation and/or other materials provided with the
**     distribution.
**   * Neither the name of The Qt Company Ltd nor the names of its
**     contributors may be used to endorse or promote products derived
**     from this software without specific prior written permission.
**
**
** THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS
** "AS IS" AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT
** LIMITED TO, THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR
** A PARTICULAR PURPOSE ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT
** OWNER OR CONTRIBUTORS BE LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL,
** SPECIAL, EXEMPLARY, OR CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT
** LIMITED TO, PROCUREMENT OF SUBSTITUTE GOODS OR SERVICES; LOSS OF USE,
** DATA, OR PROFITS; OR BUSINESS INTERRUPTION) HOWEVER CAUSED AND ON ANY
** THEORY OF LIABILITY, WHETHER IN CONTRACT, STRICT LIABILITY, OR TORT
** (INCLUDING NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY OUT OF THE USE
** OF THIS SOFTWARE, EVEN IF ADVISED OF THE POSSIBILITY OF SUCH DAMAGE."
**
** $QT_END_LICENSE$
**
****************************************************************************/

import QtQuick 2.6

Item {
    id: fpscounter
    property real fpsNow: 0
    property bool running: false
    property alias fpsVisible: fpsLabel.visible
    property int fpsInterval: 1000
    property alias color: fpsLabel.color

    Item {
        id: swapTest
        property real t
        NumberAnimation on t {
            running: fpscounter.running
            from: 0
            to: 1
            duration: fpsInterval
            loops: Animation.Infinite
        }
        onTChanged: {
            ++fpsTimer.tick
        }
    }

    Timer {
        id: fpsTimer
        running: fpscounter.running
        repeat: true
        interval: fpsInterval
        property var lastFrameTime: new Date()
        property int tick

        onTriggered: {
            var now = new Date()
            var dt = now.getTime() - lastFrameTime.getTime()
            lastFrameTime = now
            var fps = (tick * fpsInterval) / dt
            fpsNow = Math.round(fps * 10) / 10
            tick = 0

            if (fpsVisible)
                fpsLabel.updateYerself()
        }
    }

    Text {
        id: fpsLabel
        visible: false

        anchors.centerIn: parent
        font.pixelSize: 10
        color: "white"

        function updateYerself() {
            text = Math.round(fpsNow)
        }
    }
}
